

Mplus syntax files for single- and multilevel mediation models, to accompany:

Preacher, K. J., Zyphur, M. J., & Zhang, Z. (2010). A general multilevel SEM framework for assessing multilevel mediation. *Psychological Methods, 15*, 209-233.

Preacher, K. J., Zhang, Z., & Zyphur, M. J. (2011). Alternative methods for assessing mediation in multilevel data: The advantages of multilevel SEM. *Structural Equation Modeling, 18*, 161-182.

Note: In models in which the Between and Within components of a 1→1 path are estimated separately and the Within component is random, the Between component is estimated as the *contextual effect* rather than as the Between slope in Mplus (see *Mplus User's Guide*, Ex.9.2). In Examples F and J this has been addressed by adding the Within slope to the contextual effect to yield the correct Between slope component before computing the indirect effect.

A. simple mediation

TITLE: simple mediation

DATA: FILE IS mydata.dat; ! text file containing raw data in long format

VARIABLE: NAMES ARE

x m y;

USEVARIABLES ARE

x m y;

ANALYSIS: BOOTSTRAP IS 5000; ! bootstrap is recommended for simple mediation

MODEL: ! model specification follows

m ON x; ! regress mediator on independent variable

y ON x m; ! regress outcome on both mediator and independent variable

MODEL INDIRECT: ! request significance test for indirect effect of x on y via m

y IND m x; ! indirect effect of interest (ending in y and starting with x)

OUTPUT: CINTERVAL(BCBOOTSTRAP); ! request bias-corrected bootstrap

! confidence intervals

B. 2-2-1 model with latent variables (MSEM)

TITLE: 2-2-1 mediation (similar code used in example 2)

DATA: FILE IS mydata.dat; ! text file containing raw data in long format

VARIABLE: NAMES ARE

group x1 x2 x3 m1 m2 m3 m4 m5 y1 y2 y3 y4 y5;

MISSING ARE *; ! missing data denoted "*" in mydata.dat

USEVARIABLES ARE

group x1 x2 x3 m1 m2 m3 m4 m5 y1 y2 y3 y4 y5;

BETWEEN ARE x1 x2 x3 m1 m2 m3 m4 m5; ! identify variables with only Between variance;

! variables that are not claimed as "BETWEEN ARE" or "WITHIN ARE" can have

! both Within and Between variance

CLUSTER IS group; ! Level-2 grouping identifier

ANALYSIS: TYPE IS TWOLEVEL RANDOM; ! tell Mplus to perform multilevel modeling

MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
yw BY y1 y2 y3 y4 y5; ! yw is a factor defined by y1, y2, y3, y4, and y5
%BETWEEN% ! Model for Between effects follows
mb BY m1 m2 m3 m4 m5; ! mb is a factor defined by m1, m2, m3, m4, and m5
xb BY x1 x2 x3; ! xb is a factor defined by x1, x2, and x3
yb BY y1 y2 y3 y4 y5; ! yb is a factor defined by y1, y2, y3, y4, and y5
mb ON xb(a); ! regress mb on xb, call the slope "a"
yb ON mb(b); ! regress yb on mb, call the slope "b"
yb ON xb; ! regress yb on xb, too
MODEL CONSTRAINT: ! section for computing indirect effect
NEW(ab); ! name the indirect effect
ab = a*b; ! compute the indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
! optimization history, and confidence intervals for all effects

C. 2-1-1 model (traditional MLM)

TITLE: 2-1-1 mediation (traditional MLM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE
group x m y;
USEVARIABLES ARE
group x m y;
BETWEEN IS x; ! identify variables with only Between variance;
! variables that are not claimed as "BETWEEN IS" or "WITHIN IS" can have
! both Within and Between variance
CLUSTER IS group; ! Level-2 grouping identifier
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
m y; ! estimate Level-1 (residual) variances for m and y
y ON m(b); ! regress y on m, call the slope "b"
%BETWEEN% ! Model for Between effects follows
x m y; ! estimate Level-2 (residual) variances for x, m, and y
m ON x(a); ! regress m on x, call the slope "a"
y ON m(b); ! regress y on m, constrain the slope equal to "b"
y ON x; ! regress y on x
MODEL CONSTRAINT: ! section for computing indirect effect
NEW(indb); ! name the indirect effect
indb=a*b; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
! optimization history, and confidence intervals for all effects

D. 2-1-1 model (unconflated MLM)

```

TITLE: 2-1-1 mediation (unconflated MLM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE
group x m y mmean;
USEVARIABLES ARE
group x m y mmean;
BETWEEN ARE x mmean; ! identify variables with only Between variance;
      ! variables that are not claimed as "BETWEEN ARE" or "WITHIN ARE" can have
      ! both Within and Between variance
WITHIN ARE m; ! identify variables with only Within variance
CENTERING IS GROUPMEAN(m); ! group-mean center m
CLUSTER IS group; ! Level-2 grouping identifier
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
m y; ! estimate Level-1 (residual) variances for m and y
y ON m; ! regress y on m
[m@0]; ! m was group-mean centered, so fix its mean to zero
%BETWEEN% ! Model for Between effects follows
y mmean; ! estimate Level-2 (residual) variances for y and mmean
mmean ON x(a); ! regress mmean on x, call the slope "a"
y ON mmean(b); ! regress y on mmean, call the slope "b"
y ON x; ! regress y on x
MODEL CONSTRAINT: ! section for computing indirect effect
NEW(indb); ! name the indirect effect
indb=a*b; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
      ! optimization history, and confidence intervals for all effects

```

E. 2-1-1 model (MSEM)

```

TITLE: 2-1-1 mediation (MSEM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE
group x m y;
USEVARIABLES ARE
group x m y;
BETWEEN IS x; ! identify variables with only Between variance;
      ! variables that are not claimed as "BETWEEN IS" or "WITHIN IS" can have
      ! both Within and Between variance
CLUSTER IS group; ! Level-2 grouping identifier
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows

```

```

m y; ! estimate Level-1 (residual) variances for m and y
y ON m; ! regress y on m
%BETWEEN% ! Model for Between effects follows
x m y; ! estimate Level-2 (residual) variances for x, m, and y
m ON x(a); ! regress m on x, call the slope "a"
y ON m(b); ! regress y on m, call the slope "b"
y ON x; ! regress y on x
MODEL CONSTRAINT: ! section for computing indirect effect
NEW(indb); ! name the indirect effect
indb=a*b; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
! optimization history, and confidence intervals for all effects

```

F. 2-1-1 model with random slopes (MSEM)

```

TITLE: 2-1-1 mediation (MSEM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE
group x m y;
USEVARIABLES ARE
group x m y;
BETWEEN IS x; ! identify variables with only Between variance;
! variables that are not claimed as "BETWEEN IS" or "WITHIN IS" can have
! both Within and Between variance
CLUSTER IS group; ! Level-2 grouping identifier
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
m y; ! estimate Level-1 (residual) variances for m and y
sb | y ON m; ! regress y on m
%BETWEEN% ! Model for Between effects follows
x m y; ! estimate Level-2 (residual) variances for x, m, and y
m ON x(a); ! regress m on x, call the slope "a"
y ON m(bb); ! regress y on m, call the slope "bb"; bb = contextual effect, not the Between slope
y ON x; ! regress y on x
sb WITH x m y; ! estimate Level-2 covariances of sb with x, m, and y
[sb](bw); ! estimate the mean of sb, call it "bw"
MODEL CONSTRAINT: ! section for computing indirect effect
NEW(b indb); ! name the Between b path and the indirect effect
b=bb+bw; ! compute Between b path
indb=a*b; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
! optimization history, and confidence intervals for all effects

```

G. 1-1-1 model (traditional MLM)

```

TITLE: 1-1-1 mediation (traditional MLM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE
id x m y;
USEVARIABLES ARE
id x m y;
CLUSTER IS id; ! Level-2 grouping identifier
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
sa | m ON x; ! regress m on x, call the random slope "sa"
sb | y ON m; ! regress y on m, call the random slope "sb"
sc | y ON x; ! regress y on x, call the random slope "sc"
%BETWEEN% ! Model for Between effects follows
sa sb sc m y; ! estimate Level-2 (residual) variances for sa, sb, sc, m, and y
[sa](a); ! estimate the mean of sa, call it "a"
[sb](b); ! estimate the mean of sb, call it "b"
sa WITH sc m y; ! estimate Level-2 covariances of sa with sc, m, and y
sb WITH sc m y; ! estimate Level-2 covariances of sb with sc, m, and y
sc WITH m y; ! estimate Level-2 covariances of sc with m and y
y WITH m; ! estimate Level-2 covariance of y and m
sa WITH sb(cab); ! estimate Level-2 covariance of sa and sb, call it "cab"
MODEL CONSTRAINT: ! section for computing indirect effect
NEW(ind); ! name the indirect effect
ind=a*b+cab; ! compute the indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
! optimization history, and confidence intervals for all effects

```

H. 1-1-1 model (unconflated MLM)

```

TITLE: 1-1-1 mediation (unconflated MLM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE id x m y xmean mmean ymean;
USEVARIABLES ARE id x m y xmean mmean;
CENTERING IS GROUPMEAN(x m); ! group-mean center x and m
CLUSTER IS id; ! Level-2 grouping identifier
WITHIN ARE x m; ! identify variables with only Within variance;
! variables that are not claimed as "BETWEEN ARE" or "WITHIN ARE" can have
! both Within and Between variance
BETWEEN ARE xmean mmean; ! identify variables with only Between variance
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
m ON x(aw); ! regress m on x, call the slope "aw"

```

```

y ON m(bw); ! regress y on m, call the slope "bw"
y ON x; ! regress y on x
[m@0]; ! m was group-mean centered, so fix its mean to zero
%BETWEEN% ! Model for Between effects follows
mmean y; ! estimate Level-2 (residual) variances for mmean and y
mmean ON xmean (ab); ! regress mmean on xmean, call the slope "ab"
y ON mmean (bb); ! regress y on mmean, call the slope "bb"
y ON xmean; ! regress y on xmean
MODEL CONSTRAINT: ! section for computing indirect effects
NEW(indb indw); ! name the indirect effects
indw=aw*bw; ! compute the Within indirect effect
indb=ab*bb; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
      ! optimization history, and confidence intervals for all effects

```

I. 1-1-1 model with fixed slopes (MSEM)

```

TITLE: 1-1-1 mediation (MSEM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE
id x m y;
USEVARIABLES ARE
id x m y;
CLUSTER IS id; ! Level-2 grouping identifier
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
m ON x(aw); ! regress m on x, call the slope "aw"
y ON m(bw); ! regress y on m, call the slope "bw"
y ON x; ! regress y on x
%BETWEEN% ! Model for Between effects follows
x m y; ! estimate Level-2 (residual) variances for x, m, and y
m ON x(ab); ! regress m on x, call the slope "ab"
y ON m(bb); ! regress y on m, call the slope "bb"
y ON x; ! regress y on x
MODEL CONSTRAINT: ! section for computing indirect effects
NEW(indb indw); ! name the indirect effects
indw=aw*bw; ! compute the Within indirect effect
indb=ab*bb; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
      ! optimization history, and confidence intervals for all effects

```

J. 1-1-1 model with random slopes (MSEM)

```

TITLE: 1-1-1 mediation (MSEM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE
id x m y;
USEVARIABLES ARE
id x m y;
CLUSTER IS id; ! Level-2 grouping identifier
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
sa | m ON x; ! regress m on x, call the random slope "sa"
sb | y ON m; ! regress y on m, call the random slope "sb"
sc | y ON x; ! regress y on x, call the random slope "sc"
%BETWEEN% ! Model for Between effects follows
sa sb sc x m y; ! estimate Level-2 (residual) variances for sa, sb, sc, x, m, and y
sa WITH sc x m y; ! estimate Level-2 covariances of sa with sc, x, m, and y
sa WITH sb(cab); ! estimate Level-2 covariance of sa and sb, call it "cab"
sb WITH sc x m y; ! estimate Level-2 covariances of sb with sc, x, m, and y
sc WITH x m y; ! estimate Level-2 covariances of sc with x, m, and y
m ON x(ab); ! regress m on x, call the slope "ab"; ab = contextual effect, not the Between slope
y ON m(bb); ! regress y on m, call the slope "bb"; bb = contextual effect, not the Between slope
y ON x; ! regress y on x
[sa](aw); ! estimate the mean of sa, call it "aw"
[sb](bw); ! estimate the mean of sb, call it "bw"
MODEL CONSTRAINT: ! section for computing indirect effects
NEW(a b indb indw); ! name the indirect effects
a=aw+ab; ! compute Between a path
b=bw+bb; ! compute Between b path
indw=aw*bw+cab; ! compute the Within indirect effect
indb=a*b; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
! optimization history, and confidence intervals for all effects

```

K. 2-1-2 model (MSEM)

```

TITLE: 2-1-2 mediation (MSEM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE
id x m y;
USEVARIABLES ARE
id x m y;
CLUSTER IS id; ! Level-2 grouping identifier

```

```

BETWEEN ARE x y; ! identify variables with only Between variance;
      ! variables that are not claimed as "BETWEEN ARE" or "WITHIN ARE" can have
      ! both Within and Between variance
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
m; ! estimate Level-1 (residual) variance for m
%BETWEEN% ! Model for Between effects follows
x y; ! estimate Level-2 (residual) variances for x and y
m ON x(a); ! regress m on x, call the slope "a"
y ON m(b); ! regress y on m, call the slope "b"
y ON x; ! regress y on x
MODEL CONSTRAINT: ! section for computing indirect effect
NEW(indb); ! name the indirect effect
indb=a*b; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
      ! optimization history, and confidence intervals for all effects

```

L. 1-2-1 model (MSEM)

```

TITLE: 1-2-1 mediation (MSEM)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE id x m y;
USEVARIABLES ARE id x y m;
CLUSTER IS id; ! Level-2 grouping identifier
BETWEEN ARE m; ! identify variables with only Between variance;
      ! variables that are not claimed as "BETWEEN ARE" or "WITHIN ARE" can have
      ! both Within and Between variance
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
y ON x; ! regress y on x
%BETWEEN% ! Model for Between effects follows
x m y; ! estimate Level-2 (residual) variances for x, m, and y
m ON x(a); ! regress m on x, call the slope "a"
y ON m(b); ! regress y on m, call the slope "b"
y ON x; ! regress y on x
MODEL CONSTRAINT: ! section for computing indirect effect
NEW(indb); ! name the indirect effect
indb=a*b; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
      ! optimization history, and confidence intervals for all effects

```


M. 1-2-2 model (MSEM)

TITLE: 1-2-2 mediation (MSEM)
 DATA: FILE IS mydata.dat; ! text file containing raw data in long format
 VARIABLE: NAMES ARE id x m y;
 USEVARIABLES ARE id x m y;
 CLUSTER IS id; ! Level-2 grouping identifier
 BETWEEN ARE m y; ! identify variables with only Between variance;
 ! variables that are not claimed as "BETWEEN ARE" or "WITHIN ARE" can have
 ! both Within and Between variance
 ANALYSIS: TYPE IS TWOLEVEL RANDOM;
 MODEL: ! model specification follows
 %WITHIN% ! Model for Within effects follows
 x; ! estimate Level-1 (residual) variance for x
 %BETWEEN% ! Model for Between effects follows
 m y; ! estimate Level-2 (residual) variances for m and y
 m ON x(a); ! regress m on x, call the slope "a"
 y ON m(b); ! regress y on m, call the slope "b"
 y ON x; ! regress y on x
 MODEL CONSTRAINT: ! section for computing indirect effect
 NEW(indb); ! name the indirect effect
 indb=a*b; ! compute the Between indirect effect
 OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
 ! optimization history, and confidence intervals for all effects

N. 1-1-2 model with latent variables (MSEM)

TITLE: 1-1-2 mediation (similar code used in example 3)
 DATA: FILE IS mydata.dat; ! text file containing raw data in long format
 VARIABLE: NAMES ARE
 group x1 x2 x3 x4 m1 m2 m3 y1 y2 y3 y4 y5;
 MISSING ARE *; ! missing data denoted "*" in mydata.dat
 USEVARIABLES ARE
 group x1 x2 x3 x4 m1 m2 m3 y1 y2 y3 y4 y5;
 BETWEEN ARE y1 y2 y3 y4 y5; ! identify variables with only Between variance;
 ! variables that are not claimed as "BETWEEN ARE" or "WITHIN ARE" can have
 ! both Within and Between variance
 CLUSTER IS group; ! Level-2 grouping identifier
 ANALYSIS: TYPE IS TWOLEVEL RANDOM;
 MODEL: ! model specification follows
 %WITHIN% ! Model for Within effects follows
 mw BY m1 m2 m3; ! mw is a factor defined by m1, m2, and m3
 xw BY x1 x2 x3 x4; ! xw is a factor defined by x1, x2, x3, and x4
 mw ON xw; ! regress mw on xw
 %BETWEEN% ! Model for Between effects follows
 mb BY m1 m2 m3; ! mb is a factor defined by m1, m2, and m3

```

xb BY x1 x2 x3 x4; ! xb is a factor defined by x1, x2, x3, and x4
yb BY y1 y2 y3 y4 y5; ! yb is a factor defined by y1, y2, y3, y4, and y5
yb ON mb(b); ! regress yb on mb, call the slope "b"
yb ON xb; ! regress yb on xb
mb ON xb(a); ! regress mb on xb, call the slope "a"
MODEL CONSTRAINT: ! section for computing indirect effect
NEW(ab); ! name the indirect effect
ab = a*b; ! compute the Between indirect effect
OUTPUT: TECH1 TECH8 CINTERVAL; ! request parameter specifications, starting values,
      ! optimization history, and confidence intervals for all effects

```

O. 1-(1,1)-1 model with one random slope (MSEM)

```

TITLE: 1-(1,1)-1 mediation (similar code used in example 1)
DATA: FILE IS mydata.dat; ! text file containing raw data in long format
VARIABLE: NAMES ARE
group x m1 m2 y;
MISSING ARE ALL (-999); ! missing data denoted "-999" in mydata.dat
USEVARIABLES ARE
group x m1 m2 y;
CLUSTER IS group; ! Level-2 grouping identifier
ANALYSIS: TYPE IS TWOLEVEL RANDOM;
MODEL: ! model specification follows
%WITHIN% ! Model for Within effects follows
y ON m1(bw1); ! regress y on m1, call the slope "bw1"
y ON m2(bw2); ! regress y on m2, call the slope "bw2"
c | y ON x; ! regress y on x, call the random slope "c"
m1 WITH m2; ! estimate Level-1 residual covariance of m1 and m2
m1 ON x(aw1); ! regress m1 on x, call the slope "aw1"
m2 ON x(aw2); ! regress m2 on x, call the slope "aw2"
%BETWEEN% ! Model for Between effects follows
c m1 m2 y; ! estimate Level-2 (residual) variances for c, m1, m2, and y
c WITH m1 m2 y; ! estimate Level-2 (residual) covariances of c with m1, m2, and y
y ON m1(bb1); ! regress y on m1, call the slope "bb1"
y ON m2(bb2); ! regress y on m2, call the slope "bb2"
y ON x; ! regress y on x; this is the contextual effect, not the Between slope
m1 WITH m2; ! estimate Level-2 residual covariance of m1 and m2
m1 ON x(ab1); ! regress m1 on x, call the slope "ab1"
m2 ON x(ab2); ! regress m2 on x, call the slope "ab2"
[c]; ! estimate the mean of c
MODEL CONSTRAINT: ! section for computing indirect effects and contrasts
NEW(abw1 abw2 abb1 abb2 conw conb); ! name the indirect effects and contrasts
abw1 = aw1*bw1; ! compute the first Within indirect effect
abw2 = aw2*bw2; ! compute the second Within indirect effect
abb1 = ab1*bb1; ! compute the first Between indirect effect
abb2 = ab2*bb2; ! compute the second Between indirect effect

```

```
conw = abw1-abw2; ! compute the contrast of the Within indirect effects
conb = abb1-abb2; ! compute the contrast of the Between indirect effects
OUTPUT: TECH1 TECH8; ! request parameter specifications, starting values, and
! optimization history
```